Calculation of all points where Whitney's C4-cone is everything

by Rüdiger W. Braun, Reinhold Meise, and B.A. Taylor

Abstract

Let V be a hypersurface in 4-dimensional complex space. All points p in V are determined for which C4(V, p) is the entire space. This is the program referred to in Example 22 of [1].

Initialization

```
> restart:
> with(linalg):
Warning, new definition for norm
Warning, new definition for trace
> with(Groebner):
```

Definition of *f* with $V = \mathrm{V}(f)$ and introduction of variables

```
> latin := [x,y,z,w,t]: greek := [xi,eta,zeta,omega,tau]:
> aux := [kappa, lambda, mu, nu, pi]:
> f := y*(x^2 - y^2)*w^2 - y*z^3*w + z^5;
```

$$f := y\,(x^2 - y^2)\,w^2 - y\,z^3\,w + z^5$$

```
> gr := grad(f, latin);
```

$gr :=$

$$[2\,y\,x\,w^2, (x^2 - y^2)\,w^2 - 2\,y^2\,w^2 - z^3\,w, -3\,y\,z^2\,w + 5\,z^4, 2\,y\,(x^2 - y^2)\,w - y\,z^3, 0]$$

Determination of the singular set

```
> solve(convert(gr, set) union {f});
```

$$\{w = 0, z = 0, y = y, x = x\}, \{y = 0, z = 0, x = 0, w = w\}$$

The singular set consists of a line and a plane.

Programs

The following function replaces polynomials of degree at least 1 in one of the auxiliary variables by 0:

```
> remove_aux := proc(poly)
>   global aux;
>   if indets(poly) intersect convert(aux, set) = {} then
  poly
>   else 0
>   fi
> end:
```

The following function returns an expression sequence consisting of the coefficients of all monomials in ξ, η, ζ, ω. The actual work is done recursively by get_coeffs_help.

```
> get_coeffs := proc(poly)
>   global greek, c4_tmp;
>   c4_tmp := {};
>   get_coeffs_help(poly, greek);
>   op(c4_tmp);
> end:
> get_coeffs_help := proc(poly, vars)
>   global c4_tmp;
>   local p;
>   p := collect(poly, vars[1]);
>   if nops(vars) = 1 then
>     c4_tmp := c4_tmp union {coeffs(p, vars[1])};
>   else map(get_coeffs_help, {coeffs(p, vars[1])},
  [seq(vars[i], i = 2 .. nops(vars))]);
>   fi;
> end:
```

Calculation of $C_4(V)$

$C_4(V)$ is the Zariski closure of the set M of all points $[p, v]$ with p a regular point of V and v tangent to V at p. To determine the Zariski closure, one has to write M as projection of an algebraic set. This set will be $V(f, g, g_{aux})$ in $C^{(3n)}$ with g and g_{aux} given below. The projection forgets the auxiliary variables.

```
> g := evalc(dotprod(greek, gr));
```

$$g := 2\,\xi\,y\,x\,w^2 + \eta\,(w^2\,x^2 - 3\,y^2\,w^2 - z^3\,w) + \zeta\,(-3\,y\,z^2\,w + 5\,z^4) + \omega\,(2\,y\,w\,x^2 - 2\,y^3\,w - y\,z^3)$$

```
> g_aux := evalc(dotprod(aux, gr) - 1);
```

$$g\_aux := 2\,\kappa\,y\,x\,w^2 + \lambda\,(w^2\,x^2 - 3\,y^2\,w^2 - z^3\,w) + \mu\,(-3\,y\,z^2\,w + 5\,z^4) + \nu\,(2\,y\,w\,x^2 - 2\,y^3\,w - y\,z^3) - 1$$

Calculate the Gröbner basis of the ideal $\langle f, g, g_{aux} \rangle$ with respect to an elimination order to eliminate the auxiliary variables.

```
> GB := gbasis({f, g, g_aux}, lexdeg(aux, [op(latin),
  op(greek)])): nops(GB);
```

$$32$$

The elements of GB are published in a separate file.

```
> B := map(remove_aux, GB):
> map(print, convert(B, set));
```

$$y\,w^2\,x^2 - y^3\,w^2 - y\,z^3\,w + z^5$$

$$0$$

$$2\,\xi\,y\,x\,w^2 + \eta\,w^2\,x^2 - 3\,\eta\,y^2\,w^2 - \eta\,z^3\,w - 3\,\zeta\,y\,z^2\,w + 5\,\zeta\,z^4 + 2\,\omega\,y\,w\,x^2 - 2\,\omega\,y^3\,w - \omega\,y\,z^3$$

$$\omega^2\,z\,y^5 - 2\,w\,y^3\,\omega\,z\,x\,\xi + 2\,w\,y^4\,\omega\,\eta\,z + 3\,w\,y^3\,\omega\,x^2\,\zeta - 3\,w\,y^5\,\omega\,\zeta - \omega^2\,z\,y^3\,x^2$$

$$+ 4 x^4 y^2 \omega^2 + 100 x^2 y^3 \zeta^2 w - 8 x^2 y^4 \omega^2 + 4 y^6 \omega^2 + 2 y^2 \omega x \xi z^3 + 11 y^4 \omega \zeta z^2 - y^3 \omega \eta z^3 - 40 x^2 \zeta z y^2 \eta w + 10 x^4 \zeta z \eta w - x^2 y \omega z^3 \eta + 4 x^3 y^2 \omega w \xi - 40 x^2 y^3 \omega \zeta z - 8 x^2 y^3 \omega \eta w - 11 x^2 y^2 \omega z^2 \zeta + 20 x^3 \zeta \xi z y w - 10 x^2 \zeta z^4 \eta - 50 x^4 \zeta^2 y w + 20 x^4 y \omega \zeta z + 2 x^4 y \omega \eta w - 4 y^4 \omega x w \xi - 20 y^3 \zeta x \xi z w + 30 y^4 \zeta z \eta w + 10 y^2 \zeta z^4 \eta - 50 y^5 \zeta^2 w - 20 y^3 \zeta^2 z^3 + 20 y^5 \omega \zeta z + 6 y^5 \omega \eta w + 20 x^2 \zeta^2 y z^3$$

$$2 y x w \xi z^2 + z^2 \eta w x^2 - 3 z^2 \eta y^2 w - z^3 y \eta w + x^2 \eta y w^2 - y^3 \eta w^2 + 2 y \zeta z^4 - 5 x^2 w y \zeta z + 5 y^3 \zeta z w + 2 y z^2 x^2 \omega - 2 z^2 y^3 \omega - z^3 y^2 \omega + x^2 \omega y^2 w - w y^4 \omega$$

$$-w y^5 \omega + 4 z y^5 \omega - y^2 \eta z^4 + y^3 \zeta z^3 + 2 z^4 y^2 \zeta - z^2 y^4 \omega - x^2 z^4 \eta - w y^2 \eta z^3 + y^3 x^2 \omega w - y^3 \omega z^3 + 4 x^4 z y \omega + x^2 z^2 y^2 \omega + 4 x^3 \xi z y w - 8 x^2 z y^2 \eta w - 10 x^4 \zeta y w - 8 x^2 z y^3 \omega + 20 x^2 y^3 \zeta w - x^2 \zeta y z^3 + 2 x^4 z \eta w + 6 y^4 z \eta w - 2 z x^2 \zeta y^2 w + 2 z y^4 \zeta w + 2 y x \xi z^4 + z^2 x^2 y \eta w - z^2 y^3 \eta w - 4 y^3 x \xi z w - 10 y^5 \zeta w + x^2 \eta y^2 w^2 - y^4 \eta w^2$$

$$\{ \}$$

The set $C_4(V)$ is the set of common zeros of all functions in B. For particular point p, the cone $C_4(V, p)$ consists of all v with $[p, v]$ in $C_4(V)$.

Determination of all points p where $C_4(V, p)$ is everything.

Recover all coefficients (xi,eta,zeta,omega;) coming up in B:

```
> eq_C4 := convert(map(get_coeffs, B), set);
```

$$eq\_C4 := \{y w^2 x^2 - y^3 w^2 - y z^3 w + z^5, w^2 x^2 - 3 y^2 w^2 - z^3 w, 2 y w x^2 - 2 y^3 w - y z^3, 100 x^2 y^3 w + 20 x^2 y z^3 - 50 y^5 w - 20 y^3 z^3 - 50 x^4 y w, 0, z y^5 + 4 x^4 y^2 - 8 x^2 y^4 - z y^3 x^2 + 4 y^6, 2 y x w z^2, 3 x^2 y^3 w - 40 z y^3 x^2 + 20 x^4 z y - 3 y^5 w + 11 z^2 y^4 - 11 x^2 z^2 y^2 + 20 z y^5, 30 z y^4 w - 10 x^2 z^4 - 40 x^2 z y^2 w + 10 x^4 z w + 10 y^2 z^4, -x^2 y z^3 + 6 y^5 w - 8 x^2 y^3 w + 2 x^4 y w - y^3 z^3 + 2 z y^4 w, -3 y z^2 w + 5 z^4, -20 y^3 x z w + 20 x^3 z y w, -4 y^3 x z w + 2 y x z^4 + 4 x^3 z y w, 2 y x w^2, -4 y^4 x w + 2 y^2 x z^3 + 4 x^3 y^2 w - 2 y^3 x z w, -w y^4 + 2 y z^2 x^2 - z^3 y^2 + x^2 y^2 w - 2 z^2 y^3, z^2 w x^2 - y^3 w^2 - 3 z^2 y^2 w - y z^3 w + y w^2 x^2, 5 y^3 z w + 2 y z^4 - 5 x^2 w y z, -y^5 w + 4 z y^5 + 4 x^4 z y + x^2 z^2 y^2 + x^2 y^3 w - z^2 y^4 - 8 z y^3 x^2 - y^3 z^3, -8 x^2 z y^2 w + 2 x^4 z w + 6 z y^4 w - y^2 z^4 - w y^2 z^3 + z^2 x^2 y w - z^2 y^3 w - x^2 z^4 + x^2 y^2 w^2 - y^4 w^2, y^3 z^3 + 2 y^2 z^4 + 20 x^2 y^3 w + 2 z y^4 w - x^2 y z^3 - 10 x^4 y w - 2 x^2 z y^2 w - 10 y^5 w\}$$

```
> nops(eq_C4);
```

$$21$$

```
> solve(eq_C4);
```

$$\{x=x, w=0, y=0, z=0\}, \{y=y, w=0, z=0, x=y\},$$
$$\{y=y, w=0, z=0, x=-y\}, \{w=w, y=0, z=0, x=0\}$$

These 4 lines constitute the set of all points p where $C_4(V, p) = C^n$.

References

[1] R. W. Braun, R. Meise, B. A. Taylor: Surjectivity of constant coefficient partial differential operators on A(R^4) and Whitney's C_4-cone. Preprint.

[2] E. M. Chirka: Complex Analytic Sets. Kluwer, Dordrecht 1989. Translated from the Russian.

[3] D. Cox, J. Little,D. O'Shea: Ideals, Varieties, and Algorithms. Springer, New Yoork 1997.

[4] H. Whitney: Complex Analytic Varieties. Addison Wesley, Reading, Mass. 1972.

Authors's email addresses

Ruediger.Braun@uni-duesseldorf.de
meise@cs.uni-duesseldorf.de
taylor@math.lsa.umich.edu